Operacje SSE:

Sufiks **PS**



Operacje SSE – c.d.:

Sufiks **SS**



Step 1:

Compute Pi (Leibniz formula) using SSE instructions

```
        .data
        .align 16
denom:
        .double 1.0, 3.0        # first & second denominators
numer:
        .double 4.0, -4.0       # first & second numerators
add4:
        .double 4.0, 4.0        # difference between denominators
zero:
        .double 0.0, 0.0        # sums starting values
        .text
        .type fun_a, @function
        .global fun_a
```

Step 2:

```
fun_a:
        shr $1, %rdi                    # two terms are computed in parallel
        inc %rdi                        # half of iterations is enough
        movdqa  denom, %xmm5            # denominators to xmm5
        movdqa  numer, %xmm2            # numerators to xmm2
        movdqa  add4, %xmm3             # differences to xmm3
        movdqa  %xmm2, %xmm4            # numerators to xmm4
        movdqa  zero, %xmm1             # zeros to xmm1
```

xmm0		
xmm1	0.0	0.0
xmm2	4.0	-4.0
xmm3	4.0	4.0
xmm4	4.0	-4.0
xmm5	1.0	3.0

Step 3:

```
next:
divpd    %xmm5, %xmm2        # xmm2 /= xmm5
addpd    %xmm2, %xmm1        # xmm1 += xmm2
movdqa   %xmm4, %xmm2        # xmm2 = xmm4
addpd    %xmm3, %xmm5        # xmm5 += xmm3
dec %rdi
jnz next
```

xmm0		
xmm1	0.0+4/1+4/5…	0.0-4/3-4/7…
xmm2	4.0/1.0, 4/5, …	-4.0/3.0, -4/7, …
xmm3	4.0	4.0
xmm4	4.0	-4.0
xmm5	1.0, 5.0, 9.0, …	3.0, 7.0, 11.0, …

Step 4:

```
haddpd %xmm1, %xmm1      # horizontal sums of low & high parts
movsd  %xmm1, %xmm0      # low part to xmm0
ret                      # that's all
```

xmm0	?	4/1-4/3+4/5-4/7+…
xmm1	4/1-4/3+4/5-4/7…	4/1-4/3+4/5-4/7…
xmm2		
xmm3		
xmm4		
xmm5		